
;-----------------------------------------------------------------
;Przykładowy program obsługi miernika czestotliwosci CLK_zad1
;8253 pod adresem 80h-83h w osmiobitowej przestrzeni pamieci danych
;Czestotliwosc taktowania mikrokontrolera 6 Mhz
;Licznik T0 pracuje jako generator czestotliwosci wzorcowej 1 kHz
;dla kanalu 2 ukladu 8253. Kanal 2 generuje impuls podstawy czasu
;bramkujacy kanaly 0 i 1, ktore sluza do zliczania impulsow wejsciowych.
;-----------------------------------------------------------------
;Zmienne
;-----------------------------------------------------------------
  wynik_0 equ 20h ;cztery bajty na wynik pomiaru
  wynik_1 equ 21h
  wynik_2 equ 22h
  wynik_3 equ 23h
;-----------------------------------------------------------------
;Poczatek programu
;-----------------------------------------------------------------
  org 00h ;procesor rozpoczyna dzialanie od adresu 00
  sjmp start ;ominiecie wektorow przerwan
;-----------------------------------------------------------------
;Wektory przerwan
;-----------------------------------------------------------------
  org 03h
  ljmp int_X0 ;procedura obslugi przerwania zewnetrznego INT0
  org 0Bh
  ljmp int_T0 ;procedura obslugi przerwania od licznika T0
;-----------------------------------------------------------------
;Poczatek kodu
;-----------------------------------------------------------------
start:
;----------------- konfiguracja licznika T0 w 8051 ---------------
  mov TMOD,#00000010b ;0 0 0 0 0 0 1 0

;| | | | | | ---- 10 - tryb 2 (8 bit, auto reload)
;| | | | | ------ 0 – czasomierz
;| | | | -------- 0 – wylaczenie bramkowania
;---------------- 0000 – dotyczy licznika T1

  mov TH0,#6 ;licznik ma liczyc 250 impulsow (256-250=6)
  setb TR0 ;wlaczenie zliczania
;----------------- konfiguracja przerwan w 8051 ------------------
  setb EX0 ;wlaczenie przerwania zewnetrznego INT0
  setb ET0 ;wlaczenie przerwania od ukladu czasowego T0
  setb EA ;wlaczenie systemu przerwan
;----------------- konfiguracja kanalu 0 w 8253 ------------------
  mov R0,#83h ;rejestr sterujacy 8253
  mov A,#00110100b ;0 0 1 1 0 1 0 0

;| | | | | | | -- 0 – licznik binarny; 1 – BCD
;| | | | -------- 010 – tryb 2
;| | ------------ 11 – stala podawana na dwoch bajtach
;---------------- 00 – numer kanalu (kanal 0)

  movx @R0,A
  mov R0,#80h ;adres rejestru licznika numer 0
  mov A,#FFh ;stala w kanale 0 licznika 8253 (FFFFh)
  movx @R0,A ;najpierw mlodszy
  movx @R0,A ;potem starszy bajt



;----------------- konfiguracja kanalu 1 w 8253 ------------------
  mov R0,#83h ;rejestr sterujacy 8253
  mov A,#01110000b ;0 1 1 1 0 0 0 0

;| | | | | | | -- 0 – licznik binarny; 1 – BCD
;| | | | -------- 000 – tryb 0
;| | ------------ 11 – stala podawana na dwoch bajtach
;---------------- 01 – numer kanalu (kanal 1)

  movx @R0,A
  mov R0,#81h ;adres rejestru licznika numer 0
  mov A,#FFh ;stala w kanale 1 licznika 8253 (FFFFh)
  movx @R0,A ;najpierw mlodszy
  movx @R0,A ;potem starszy bajt
;----------------- konfiguracja kanalu 2 w 8253 ------------------
  mov R0,#83h ;rejestr sterujacy 8253
  mov A,#10110000b ;1 0 1 1 0 0 0 0

;| | | | | | | -- 0 – licznik binarny; 1 – BCD
;| | | | -------- 000 – tryb 0
;| | ------------ 11 – stala podawana na dwoch bajtach
;---------------- 10 – numer kanalu (kanal 2)

  movx @R0,A
;----------------- ten fragment programu startuje pomiar ---------
  mov R0,#82h ;adres rejestru licznika numer 0
  mov A,#232 ;stala w kanale 2 licznika 8253 (1000d)
  movx @R0,A ;najpierw mlodszy
  mov A,#3
  movx @R0,A ;potem starszy bajt
;----------------- koniec fragmentu programu do startu pomiaru ---
;...
;-----------------------------------------------------------------
;Przerwanie z wejscia INT0 zglaszane po zakonczeniu pomiaru
;-----------------------------------------------------------------
int_X0:
  push ACC
  push PSW
  jb P1.0,int_X0_blad ;jesli nastapilo przepelnienie to obsluga bledu
  mov R0,#80h ;adres zerowego kanalu licznika 8253
  movx A,@R0 ;odczyt najmlodszego bajtu (pierwszy z czterech)
  cpl A ;wynikiem jest dopelnienie dwojkowe
  add A,#1
  mov wynik_0,A ;najmlodszy bajt wyniku
  movx A,@R0 ;odczyt mlodszego bajtu (drugi z czterech)
  cpl A ;wynikiem jest dopelnienie dwojkowe
  addc A,#0 ;dodanie przeniesienia z poprzedniego bajtu
  mov wynik_1,A ;mlodszy bajt wyniku
  mov R0,#81h ;adres pierwszego kanalu licznika 8253
  movx A,@R0 ;odczyt starszego bajtu (trzeci z czterech)
  cpl A ;wynikiem jest dopelnienie dwojkowe
  addc A,#0 ;dodanie przeniesienia z poprzedniego bajtu
  mov wynik_2,A ;najmlodszy bajt wyniku
  movx A,@R0 ;odczyt najstarszego bajtu (ostatni z czterech)
  cpl A ;wynikiem jest dopelnienie dwojkowe
  addc A,#0 ;dodanie przeniesienia z poprzedniego bajtu
  mov wynik_3,A ;mlodszy bajt wyniku
  pop PSW
  pop ACC
  reti



;-----------------------------------------------------------------
;Przerwanie od ukladu czasowo-licznikowego zglaszane
;z czestotliwoscia 2kHz
;-----------------------------------------------------------------
int_T0:
  cpl P3.4 ;generowanie fali prostokatnej o czestotliwosci 1kHz
  reti
;-----------------------------------------------------------------
;----------------- koniec ----------------------------------------
;-----------------------------------------------------------------
;Objasnienia trybow ukladow czasowo-licznikowych stosowanych w ukladzie:
;Tryb 2 licznika T0 w 8051
; - impulsy sa zliczane w 8-bitowym rejestrze TH0 (w gore)
; - w rejestrze TL0 jest stala automatycznie przepisywana przy przejsciu FF-00
; - przy przepelnieniu jest takze generowane przerwanie
;Tryb 0 kanalow 1 i 2 ukladu 8253
; - impulsy sa zliczane w 16-bitowym liczniku (w dol) jesli GATE = 1
; - rozpoczecie zliczania jest inicjowane programowo porzez zapis 16-bitowej
;   stalej (w porcjach po bajcie, najpierw LSB potem MSB)
; - po zakonczeniu zliczania (w chwili gdy licznik osiagnal 0)
;   podawana jest na stale ”1” na wyjscie OUT
;Tryb 2 kanalu 0 ukladu 8253
; - impulsy sa zliczane w 16-bitowym liczniku (w dol) jesli GATE = 1
; - rozpoczecie zliczania inicjowane narastajacym zboczem na GATE
; - przy przejsciu z 0000 na FFFF podawany jest krotki impuls na wyjscie OUT
;-----------------------------------------------------------------
;-----------------------------------------------------------------


